
//Input: Thumb Joystick
#define TJGND A0              //
#define TJSEL A1              //
#define TJHORZ A2             //
#define TJVERT A3             //
#define TJVCC A4              //

//Output: Matrix Made EZ
#define SHIFT_DATA 8          //
#define SHIFT_CLOCK 7         //
#define SHIFT_LATCH 6         //
#define PIN_DIM 3             //
#define DEBUG false           //

uint8_t matrix[8];            //buffered image of the MatrixMadeEZ display 
volatile boolean triggered = false;   //holds the ISR state

void setup() {
  if (DEBUG) {                //conditional support for the Serial Monitor
    Serial.begin(9600);
    while (!Serial);
  }
  //Input
  pinMode(TJGND, OUTPUT);
  pinMode(TJVCC, OUTPUT);
  digitalWrite(TJGND, LOW);
  digitalWrite(TJVCC, HIGH);
  //Output
  pinMode(TJSEL, INPUT_PULLUP);
  pinMode(SHIFT_DATA, OUTPUT);
  pinMode(SHIFT_CLOCK, OUTPUT);
  pinMode(SHIFT_LATCH, OUTPUT);
  pinMode(PIN_DIM, OUTPUT);
  digitalWrite(PIN_DIM, LOW);
  // Use of the external interrupt to trigger the clearing of the display
  attachInterrupt(digitalPinToInterrupt(2), ISRClear, FALLING);
}
//Interrupt Service Routine for the Joystick's Button press
void ISRClear() {
  triggered = true;
}


void loop() {
  if (triggered) {        //clear the display if requested 
    triggered = false;    //prepare for the next interrupt
    for (uint8_t r = 0; r < 8; r++)   //cleat the display
      matrix[r] = 0;
    delay(300);           //pause to allow to settle
  }

  //map joystick reading to a viable column number
  uint8_t column = analogRead(TJHORZ) >> 7;
  //map joystick reading to a viable row number
  uint8_t row = analogRead(TJVERT) >> 7;
     
  if (DEBUG) {
    Serial.print(column);
    Serial.print('\t');
    Serial.println(row);
  }

  matrix[row] |= 1 << column;               //Set the column  (or equals)
  for (uint8_t r = 0; r < 8; r++) {         //row scanning update of display
    digitalWrite(SHIFT_LATCH, LOW);
    shiftOut(SHIFT_DATA, SHIFT_CLOCK, LSBFIRST, matrix[r]);
    shiftOut(SHIFT_DATA, SHIFT_CLOCK, MSBFIRST, 1 << (7 - r));
    digitalWrite(SHIFT_LATCH, HIGH);
  }
}

